Retour au site web

Exercice 2-3 : Développement limité de la fonction arctg(x)

a) Selon un théorème classique sur les séries à termes de signe alterné, l'erreur de troncation est majorée par le premier terme négligé (en valeur absolue).

b) Le terme général de la série est :

$$u_k = (-1)^k \frac{x^{2k+1}}{2k+1}; \qquad k = 0,1,2,\ldots$$

Si on conserve n termes (de 0 à $n-1$), le premier terme négligé correspondra à $k = n$. Pour que l'erreur sur π soit de 10^{-5}, il faut que l'erreur de troncation pour $x = 1$ soit quatre fois plus petite, ce qui détermine n :

$$\frac{1}{4}10^{-5} = \frac{1}{2n+1},$$

soit $n \simeq 2 \times 10^5$.

c) La relation

$$\text{arctg}\, x + \text{arctg}\, 1/x = \frac{\pi}{2}$$

permet d'utiliser un argument compris dans le domaine de convergence de la série. Le programme ci-dessous utilise cette remarque pour calculer numériquement arctg x.

```
//Exercice 2-3: arctg(x) par le developpement en serie

eps = 0.00001; nmax = 100000;
u = 1.0; somme = 0.0; n = 0; gd = %f;
signe = 1;

z = input("valeur de l''argument: ");
if z < 0
  z = -z; signe = -1;
end
if z > 1
  gd = %t; z = 1/z;
end
zz = z*z;
while abs(u) > eps & n < nmax
  somme = somme + u;
  u = - zz*(2*n+1)*u/(2*n+3);
  n = n+1;
  if modulo(n,100) == 0, printf("%d  %f\n",n, u), end
end
rslt = z*somme;
if gd
  rslt = %pi/2 - rslt;
  z = 1/z;
end
z = signe*z; rslt = signe*rslt;
printf("\nmon arc tg( %6.3f) = %12.8f  valeur exacte %12.8f\n",..
        z,rslt,atan(z))
```